
An RCL circuit, connected to a generator, has the following component values R=100 ohms, L =1.H  , C=5E-6 F. The driving voltage is V= 100 sin (2π f t)  , with   f= 79.6Hz.
The steady state analysis gives 

XL= 500 ohms , XC = 400 ohms  , Z= 100 + j ( 100)

or Z = 141.4 at 45 deg.

The steady state analyzed using phasors  gives,
 I =V/Z = 0.707 at (-45 deg) 

VR= IZ =70.7 at(-45 deg)

VC=(400 at(-90) ) * (.707 at (-45deg)) = 282 at (-135 deg)

VL=(500 at (+90deg) ) * ( .708 at(-45) ) = 353 at (45 deg) 

It is instructuive to solve the problem numerically ( see Fortran code below). The one sees the solution developing in time and has a sense of how much one has to wait for the transient effects to die out.

We use two Taylor series.

The differential equation is

      di/dt=  (1./L)*(-R*i -q/C +V(t) ) ≡ g (i,q,t)    .               (1a)

or  

d2q /dt2 = (1./L)*(-R*(dq/dt) -q/C +V(t) )                       (1b)

To solve numerically eq. 1b the initial conditions q(0) and (dq/dt)0
We will also need 

d2 i/dt2 = dg/dt = (1./L)*(-R*di/dt -i/C +dV(t)/dt )                (2)
The series for the charge to second order is

qn=qn-1 + dt*(dq/dt)n-1 +.5*dt**2*(d2q/dt2)n-1                     (3)             

and that for the current is

 in=in-1 + dt*(di/dt)n-1 +.5*dt**2*(d2i/dt2)n-1                        (4)   
Using ( 1-a) and (2) , (3) and (4) are rewritten as

  qn=qn-1 + dt*in-1 +.5*dt**2*gn-1                                            (5)

  in=in-1 + dt* g n-1 +.5*dt**2*(dg/dt)n-1                                  (6)   

Let  τ be the period of the generator  = 2π/ω.The plot show the voltages for the first 3(360 degrees)=  1080 degrees.
The maximum voltages should de be compared to those of the steady state regime calculated above.


[image: image1]
First oscillations for a time            0    ≤ time ≤ 3τ


[image: image2]
The steady state regime has been achieved  sfter 6 oscillations.
 The plot is for        6τ    ≤ time ≤ 8τ .
FORTRAN code

c AC circuit steady state  Schaum's BEE prob 3.7

c        L di/dt= -RI -q/c  +V(t)

      real L ,i ,i0,i1

      data Vm,f,R,L,C/100.,79.6,100.,1.,5.E-6/

      v(t)=Vm*sin(omega*t)

      g(i,q,t)= (1./L)*(-R*i -q/C +V(t) )

      dgdt(i,q,t)= (1./L)*(-R*g(i,q,t)-i/C+ omega*vm*cos(omega*t) )

      pi=2.*asin(1.)

      omega=2.*pi*f

      t1=L/R

      t2=R*C

      tau=2.*pi/omega

      tscale=amin1(t1,t2,tau)

      dt=tscale/400.

      tfinal=8.*tau

      print*,'tfinal/tL ,tfinal/Tc=', tfinal/t1, tfinal/t2

      nstep=int(tfinal/dt)

      kprint=int(float(nstep)/200.)

      kount=kprint

      q0=0.

      i0=0.

      vr0=R*i0

      vc0=q0/c

      vl0=L* g(i0,q0,0.)

      print*,'      theta (deg) ,     , Vr,     ,vc,    ,vL,

     $ ,Vsource'

      print*,'       '

      print 100,0.,vr0,vc0,vl0,v(0.)

      do 10 j=1,nstep

      t=dt*float(j)

      q1=q0 + dt*i0+.5*dt**2*g(i0,q0,t-dt)

      i1=i0 + dt*g(i0,q0,t-dt) +.5*dt**2*dgdt(i0,q0,t-dt)

      if(j.eq.kount)then

      vr=R*i1

      vc=q1/C

      vL=L*g(i1,q1,t)

      if(t.le.3.*tau)then

      print 100, omega*t*180./pi,vr,vc,vl, v(t)

      endif

      kount=kount+kprint

      endif

      q0=q1

      i0=i1

10    continue

100   format(5(3x,e11.4))

      stop

      end

RCL circuit with sinusoidal input

RCL first oscillations

